

Programmatic Challenges of Dose Tapering Using SAS®

Iuliana Barbalau, Santen Inc., Emeryville, CA

Chen Shi, Santen Inc., Emeryville, CA

Yang Yang, Santen Inc., Emeryville, CA

ABSTRACT:

In a good clinical study, statisticians and various stakeholders are interested in assessing and isolating the effect of non-study drugs. One common practice in clinical trials is that clinical investigators follow the protocol to taper certain concomitant medications in an attempt to prevent or resolve adverse reactions and/or to minimize the number of subject withdrawals due to lack of efficacy or adverse event. To assess the impact of those tapering medicines during study is of high interest to clinical scientists and the study statistician. This paper presents the challenges and caveats of assessing the impact of tapering a certain type of concomitant medications using SAS® v9.3 based on a hypothetical case. The paper also presents the advantages of visual graphs in facilitating communications between clinical scientists and the study statistician.

OUTLINE:

- Introduction
- Possible challenges and caveats of tapering
- Standardize a medication so that comparison is meaningful
- SAS® steps for achieving standardization
- Graphs of tapering medication (step wise graph)
- Conclusions

INTRODUCTION:

The focus of this paper will be on presenting challenging situations for assessing the impact of tapering medication and possible solutions using SAS® software. The tapering of a medication allows a patient's organism to gradually discontinue a specific medication and reduce the effect an abrupt stop would have. For this article purpose, we will consider Prednisone as our intended tapering medication. Prednisone is a corticosteroid¹. There are medical reasons why it is recommended gradual tapering of Prednisone. Withdrawal symptoms that could have an impact on subjects are: joint pain, muscle pain, fatigue, headache, fever, low blood pressure, nausea and vomiting. Also, an abrupt discontinuation of treatment in patients who have been on steroid for a prolonged period of time may cause severe symptoms (adrenal crisis, exogenous adrenal insufficiency) due to the fact the normal production of steroids by the body has been turned off². Let's assume for our fictive tapering plan, the tapering regimes should be as follows: 10 mg increments per week when the daily dose is more than 40 mg prednisone, 5 mg increments per week when the Prednisone daily dose is less than 40 mg of prednisone up to 20 mg of prednisone and 1 mg weekly increments afterward's³.

The data presented in this article is a fictive one and is not related to an actual study. We will present concomitant medications (steroids) taken by two subjects versus the specified tapering scheme presented in clinical study protocol. The variables included in the fictive dataset follow CDISC standards and are self-explanatory such as SUBJID (subject ID), CMSEQ (concomitant medication sequence), CMDECOD (concomitant medication coded term), CMDOSTXT (concomitant medication dose), CMDOSU (concomitant medication unit), CMDOSFRQ (concomitant medication frequency), CMROUTE (concomitant medication route), CMSTDT (concomitant

¹ <http://www.drugs.com/prednisone.html>

^{2,3} <http://arthritis.about.com/od/prednisone/f/withdrawaltaper.htm>

medication start date), CMSTDY (start day of concomitant medication), CMENDTC (concomitant medication end date), CMENDY (stop day of concomitant medication).

Table 1: CM domain (two subjects included)

SUBJID	CMSEQ	CMDECOD	CMDO STXT	CMDOSU	CMDOS -FRQ	CM-ROUTE	CMSTDTC	CMSTDY	CMENDTC	CMENDY
US100-0001	9	PREDNISOLONE	70	mg	QD	ORAL	6/4/2012	21	6/10/2012	27
US100-0001	10	PREDNISOLONE	60	mg	QD	ORAL	6/11/2012	28	6/17/2012	34
US100-0001	11	PREDNISOLONE	50	mg	QD	ORAL	6/18/2012	35	6/28/2012	45
US100-0001	12	PREDNISOLONE	40	mg	QD	ORAL	6/29/2012	46	7/4/2012	51
US100-0001	13	PREDNISOLONE	30	mg	QD	ORAL	7/5/2012	52	7/11/2012	58
US100-0001	16	PREDNISOLONE	20	mg	QD	ORAL	7/12/2012	59	7/18/2012	65
US100-0001	17	PREDNISOLONE	15	mg	QD	ORAL	7/19/2012	66	7/25/2012	72
US100-0001	18	PREDNISOLONE	10	mg	QD	ORAL	7/26/2012	73	8/6/2012	84
US100-0002	2	PREDNISONE	20	mg	QD	ORAL	1/25/2010	-1100	1/29/2013	1
US100-0002	9	PREDNISONE	12.5	mg	QD	ORAL	1/30/2013	2	2/6/2013	9
US100-0002	10	PREDNISONE	10	mg	QD	ORAL	2/7/2013	10	2/14/2013	17
US100-0002	11	PREDNISONE	7.5	mg	QD	ORAL	2/15/2013	18	2/22/2013	25
US100-0002	12	PREDNISONE	5	mg	QD	ORAL	2/23/2013	26	3/2/2013	33
US100-0002	14	PREDNISONE	2.5	mg	QD	ORAL	3/3/2013	34	3/10/2013	41
US100-0002	32	PREDNISONE	20	mg	QD	ORAL	6/26/2013	149	6/26/2013	149

POSSIBLE CHALLENGES AND CAVEATS OF TAPERING:

Challenges of tapering medication:

1. Data might be difficult to handle especially in an international clinical trial setting even if the instructions were properly explained and sites have been trained on data collection. Some of the information from the database could be confusing or hard to interpret such as routes, frequencies of medication due to multiple possible ways to prescribe a medication. Language barriers could cause problems with respect to the accuracy of data collection.
2. When the medication is overlapping, the calculation of the daily dose is challenging. There is a need to create a record for each study day and then sum per day the total daily dose.
3. In order to find a meaningful standardization of the medication, a team of people including medical doctors, scientists, statisticians and programmers need to meet several times in order to create a logical algorithm for conversion. The rules for conversions need to be accepted and used across departments by projects leaders and team members.

Potential solutions:

1. One possible solution for dataset entry issues would be to send queries to the site to confirm the data is accurate or to correct the data as necessary. This task could be very challenging and would require a lot of manual work.
2. Another option is to add edit checks to verify overlapping medications over specific periods of time, to verify the accuracy of frequency and routes of medication. Unfortunately, if there is a free-text option, an edit check might be hard to implement.

3. Creating a tapering flowchart to be used during the meetings, and to be referenced for converting medication to a 'standard' one. Please see an example of such as tapering flowchart on next page of this article.
4. If tapering of medication is specified in the clinical trial protocol, the graph of each subject by site could be used as a tool for assessing the performance of the sites. The sites should be following the protocol instructions with respect to the tapering of medication.

STANDARDIZE A MEDICATION SO THAT COMPARISON IS MEANINGFUL:

Subjects enrolled in the study might take medications that need to be tapered. Some of concomitant medications used during clinical trial could be stronger than other medications. To have a meaningful comparison, a standardization of medication is needed. Usually, this type of standardization includes a conversion factor of medication to a 'standard' medication (Prednisone in our case), a conversion of units to standard units. It is highly recommended to use a flowchart as presented in next page. The scheme presented on following page is a useful visual tool for the entire working team, not just for SAS programmers, but also for data management and clinical operations. As per protocol, we need to follow a step down tapering of the level of Prednisone a subject is taking during a period of time. The 'standard medication' makes the comparison between subjects more manageable. For example, one subject could take 1 mg of Meprednisone, which is 1.25 stronger than Prednisone, while another subject is taking only 1 mg of Prednisone. Therefore first subject will have 1.25 mg of Prednisone in his/her body while the second subject will have 1 mg of Prednisone in body at a specific timing.


SAS STEPS FOR ACHIEVING STANDARDIZATION:

Step 1: In order to achieve standardization, we will make use of formats to help us derive the conversion factors, conversion units and frequencies factors. All these conversions (see the standardization flowchart on page four of this article) are useful to create a common standard medication – used for comparison purposes.

Source Code Sample 1:

```
/* add formats */
proc format;
  invalue cmdecod
    'DEFLAZACORT'= 0.83
    'HYDROCORTISONE'= 0.25
    'MEPREDNISONE' = 1.25
    'METHYLPREDNISOLONE' = 1.25
    'PREDNISOLONE' = 1
    'PREDNISONE' = 1
    'AZATHIOPRINE' = 1
  ;
  invalue cmdosu
    'g' = 1000
    'ug' = 0.001
    'mg' = 1
  ;
  invalue cmdosfrq
    'QD' = 1
    'TID' = 3
    'BID' = 2
    'QID' = 4
  ;
run;
```

Useful Flowchart:



Step 2: In order to get the equivalent daily dose, we can calculate the standard tapering dose by multiplying the conversion factor for each standard medication with conversion factor for dosing unit and with the numeric representation of daily frequencies. Please keep in mind what kind of standard dose you are planning to calculate. In this example, we are calculating the total daily dose in mg. Some clinical trials might be interested in measuring total weekly or monthly doses. In these cases, one needs to be careful of the formats used for frequencies. The conversion applied are based on the way the data is collected (presented in clinical report forms) and also based on the type of requests received from diverse stake holders.

Source Code Sample 2:

```
/* standardize the medication dose based on flowchart */
data daily_dose1 ;
  set cm_steroids;
  where cmstdy >= 1;
  by subjid cmstdy ;
  standard_dose = input(cmdecod, cmdecod.) * input(cmdosu, cmdosu.) *
                  input(cmdosfrq, cmdosfrq.) * cmdostxt;
  do i = cmstdy to cmendy by 1 ;
    if first.cmstdy then studyday = i;
    output;
  end;
  keep subjid standard_dose studyday cmstdy cmendy;
run;
```

Step 3: We need to calculate the daily dose for each day of the interval. For example, if one subject (US100-0001) takes PREDNISOLONE (70 mg) from study day 21 until study day 27, then 7 separate records for day 21, day 22, up to day 27 are created accordingly. These records will help us obtain the correct total daily by summation. A medication is overlapping when a subject is taking different doses of Prednisone during the same time interval.

Source Code Sample 3:

```
proc sql;
  /* create total daily dose by adding together the mg of medication per day */
  create table daily_dose2
  as select unique subjid, min(cmstdy) as min_day, max(cmendy) as max_day
  from daily_dose1
  group by subjid;

  /* create a matrix to be used for figure */
  create table dose_figure
  as select unique subjid, studyday, sum(standard_dose) as total_daily_dose
  from daily_dose1
  group by subjid, studyday;
quit;

data dose_figure_final;
  merge dose_figure(in=a) daily_dose1(in=b)
        daily_dose2(in=c rename=(min_day=studyday));
  by subjid studyday;
  if c;
run;
```

Step 4: First, we are creating a 'standard' tapering dose using 10 mg increments (dataset taper10). We will use this dataset to calculate the following steps of tapering such as 5 mg and respectively 1 mg increments. In the last step, we need to merge the total daily dose – standard format with tapering doses using 10 mg, 5 mg and 1 mg increments, so that we can populate the correct tapering dose for each of the 7 days interval.

Source Code Sample 4:

```
/*create final dataset for figure*/
data taper10;
  set dose_figure_final;
  by subjid;
  if first.subjid then do;
    cnt = -1;
    do day = studyday to max_day by 7;
      cnt + 1;
      if total_daily_dose > 40 then dose10 = total_daily_dose - 10 * cnt;
      else if 20 < total_daily_dose <= 40 then dose10 = total_daily_dose - 5 * cnt;
      else if 0 <= total_daily_dose <= 20 then dose10 = total_daily_dose - 1 * cnt;
      output;
    end;
  end;
  if day <= max_day;
  drop studyday total_daily_dose cnt;
run;
proc sort data = taper10; by subjid day; run;

%macro taper(cond =, no =, increment = );
data taper&increment. (rename = (j = day));
  set taper&no. (&cond);
  by subjid day;
  if first.subjid then do;
    cnt = 0;
    do j = day to max_day by 7;
      cnt + 1;
      dose&increment. = dose&no. - &increment. * cnt;
      output;
    end;
  end;
  keep subjid j dose&increment. max_day;
run;
proc sort data=taper&increment.; by subjid day; run;
%mend taper;
%taper(cond=%str(where=(20 < dose10 <= 40)), no=10, increment=5);
%taper(cond=%str(where=(0 <= dose5 <= 20)), no=5, increment=1);

data final ;
  merge dose_figure(in = a rename = (studyday = day)) taper10 taper5 taper1;
  by subjid day;;
  if a;
  retain taper_dose;
  if cmstdy ne . then taper_dose=max(of dose10, dose5, dose1);
  keep subjid day total_daily_dose taper_dose;
  label total_daily_dose = "Daily Standard Dose (mg)"
        taper_dose = "Daily Taper Dose (mg)";
run;
```

Step 5: Output graph for each subject. First, we are using PROC SQL to create unique counters for subjects. Then we are using DO loops to create a graph of tapering medication for each subject using PROC SGPLOT.

Source Code Sample 5:

```
ods listing close;
options orientation = landscape;

%macro taper_graph;
proc sql;
  select unique subjid into: subj1 -:subj1000
```

```

from final;
select count(unique subjid) into: cnt from final;
quit;
%put &subjl &cnt;

%do i = 1 %to &cnt;
ods rtf file = "C:\Users\Ibarbalau\Desktop\Subject_&subj&i...rtf";
title "Steroids Tapering for Subject &subj&i.";
proc sgplot data = final (where = (subjid = "&subj&i."));
step x = day y = total_daily_dose/justify = center lineattrs = (pattern = solid
                                                                    thickness = 2 color = red);
step x = day y = taper_dose/ justify = center lineattrs = (pattern = dot
                                                                    thickness = 2 color = blue);
axis min = 0 max = 150 values = (0 to 150 by 7) valueshint
label = "Study Day (Steroid Start Date - First Dose Date)";
yaxis min = 0 max = 100 values = (0 to 100 by 5) integer grid
label = "Total Daily Dose (mg)";
run;
footnote1 justify=left "Dataset location:\Ibarbalau\Desktop\dose_figure.sas7bdat";
footnote2 justify=left "Program location:\Ibarbalau\Desktop\cm_tapering.sas";
%end;
ods rtf close;
%mend taper_graph;
%taper_graph;
ods listing;

```




Figure 1: Prednisone Tapering for Subject US100-0001

In the figure above, subject US100-0001, is tapering up to day 42 in concordance with our fictive protocol specifications (see introduction). Starting from day 42, there is gap between the recommended tapering dose and the actual dose taken by the subject. If the study reviewer considers the difference to be of clinical significance, then the subject needs to be further investigated.




Figure 2: Prednisone Tapering for Subject US100-0002

In the figure above, we notice the subject US100-0002 follows the tapering scheme up to day 98. After day 98, there is a 'jump' in the actual dose the subject is taking. This phenomenon is called "rescue" of the subject. Therefore, the subject is not tapering the steroids, but is rescued by being prescribed a stronger daily dose of Prednisone (mg) than the previous daily dosage.

CONCLUSIONS:

Assessing dose tapering in clinical trials poses several challenges, such as not all the medications could be "standardized" to a common denominator, difficulty in finding a consensus regarding the standardization. We can take for example the psychiatry/psychology clinical trials for which the tapering of depression or anxiety medications is hard to standardize due to the unique cognitive functioning of subject's brain. We leave such investigations for future work. In any good clinical trials, being able to standardize the medications and being able to indicate visually the tapering of specific medication, is a good tool to be utilized by Biometrics department when collaborating with medical doctors, clinical and data management departments. We heard it many times but it is all true "a picture is worth a thousand words". Clinical trials make no exception to this rule, especially when SAS programmers are trying to indicate tapering of medication to departments which don't deal with metadata on daily basis.

REFERENCES:

1. <http://arthritis.about.com/od/prednisone/f/withdrawaltaper.htm>
2. <http://www.drugs.com/prednisone.html>

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the authors at:

Name: Iuliana Barbalau
Organization: Santen Inc.
Address: 2100 Powell Street
City, State ZIP: Emeryville, CA 94608
Work Phone: 415-268-9173
Email: ibarbalau@santeninc.com
Web: www.santeninc.com

Name: Chen Shi
Organization: Santen Inc.
Address: 2100 Powell Street
City, State ZIP: Emeryville, CA 94608
Work Phone: 415-268-9178
Email: hshi@santeninc.com
Web: www.santeninc.com

Name: Yang Yang
Organization: Santen Inc.
Address: 2100 Powell Street
City, State ZIP: Emeryville, CA 94608
Work Phone: 415-268-9140
Email: yyang@santeninc.com
Web: www.santeninc.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.